
CPAN modules
Comprehensive Perl Archive Network.

Decentralized repository of modules

If you have a coding problem, someone's 
probably already handled it.

Databases, XML, CGI, templating, file access, 
parsing, networking, debugging, natural 
language, etc etc etc 



Finding modules
CPAN is the distribution channel.

Currently 6,000 modules

http://cpan.perl.org

http://search.cpan.org



Installing a module
Get the module and unpack it

wget http://search.cpan.org/....
tar zxvf distro.tar.gz
cd distro

The magic incantation
perl Makefile.PL
make
make test
sudo make install

Let's see that in action...



Now we can use it

Get a page with Mech
use WWW::Mechanize;
my $m = WWW::Mechanize->new();
$m->get( "http://plainblack.com/" );
print $m->title, "\n";

$ ./fetch.pl
Plain Black, makers of WebGUI - Home



That's where 
we're going



References
Scalars that refer to other data.

Any data item can be referred to.

Similar to pointers in C, but safer.



References
Backslash is the reference operator

my $score = 12;
my $ref = \$score;
print $ref, "\n";
print $$ref, "\n";
print $score, "\n";

$ perl ref.pl
SCALAR(0x806f24)
12
12



Array references
Reference anything with its sigil

my @stooges = qw( Larry Moe Curly );
my $ref = \@stooges;
print join( ", ", @$ref );

Larry, Moe, Curly



Hash references
Same as arrays

my %names = (
    Moe => "Howard",
    Larry => "Fine",
    Iggy => "Pop",
);
my $ref = \%names;



Dereferencing 
elements

Use an arrow, and $ for the sigil
my $href = \%names;
print $href->{Moe};

my $aref = \@list;
print $aref->[2];



Anonymous arrays
Sometimes you want a ref immediately

my @list = qw( Larry Moe Curly );
my $stooges = \@list;

Make the reference anonymous
my $stooges = [ qw( Larry Moe Curly ) ];
-or-
my $stooges = [ 'Larry','Moe','Curly' ];

No need for an intermediate array variables
The array itself has no name.



Anonymous hashes
Same as with anonymous arrays

my %names = ( Larry => 'Fine', ... );
my $stooges = \%names;

Make the reference anonymous
my $stooges = { Larry => 'Fine', ... );
print $stooges->{Larry};

No need for an intermediate hashes variables



Structures
Structures are typically anonymous hashes

my $user = {
    id => 'mhoward',
    firstname => 'Moe',
    lastname => 'Howard',
    email => 'moe@stooge.net',
};

print $user->{id};



Nested structures
Hash values are scalars, which can also be refs

my $user = {
    id => 'mhoward',
    firstname => 'Moe',
    lastname => 'Howard',
    email => 'moe@stooge.net',
    partners => [ 'lfine', 'cneal' ],
};

my @numbskulls = @{$user->{partners}};
print $numbskulls[0]; #1 numbskull

print $user->{partners}->[0]; #another way



Circular structures
Can even refer to other references

my $larry = { id => 'lfine', ... };
my $curly = { id => 'curly', ... };
my $moe = {
    id => 'mhoward',
    partners => [ $larry, $curly ],
};

print $moe->{partners}[0]{id}; # prints "lfine"

But beware of memory leaks!



Data::Dumper
All these references can get hard to read...

use Data::Dumper;
print Dumper( $stooge );

$VAR1 = {
          'email' => 'moe@stooge.net',
          'firstname' => 'Moe',
          'lastname' => 'Howard',
          'id' => 'mhoward',
          'partners' => [
                          'lfine',
                          'cneal'
                        ]
        };



Take a breath



Reusable code
Define your own subroutines.

sub underlined {
    my $msg = shift;
    my $lines = $msg;
    $lines =~ s/\S/-/g;
    return "$msg\n$lines\n";
}

print underline( "WebGUI Conference 2004" );

WebGUI Conference 2004
------ ---------- ----



Define a library
Create underline.pl

sub underlined {
    my $msg = shift;
    my $lines = $msg;
    $lines =~ s/\S/-/g;
    return "$msg\n$lines\n";
}

1;



Use the library
Create test.pl

#!/usr/bin/perl -w

use strict;
require "underline.pl";

print underlined( "WebGUI Conference 2004" );



Packages
A package is a namespace to group vars & subs
Your package may want to hide its variables

package Underline;
our $underline = "-";
sub underlined {
    my $msg = shift;
    my $lines = $msg;
    $lines =~ s/\S/$underline/g;
    return "$msg\n$lines\n";
}
1; # end of Underline.pm



Using a package
Use the package

use Underline;
$Underline::underliner = "*";
print Underline::underlined( "WebGUI Conference 2004" );

WebGUI Conference 2004
****** ********** ****

It works, but what a pain.



Exporting
Make your module export some symbols

package Underline;
use Exporter;
our @ISA = qw( Exporter );
our @EXPORT = qw( $underline underlined );
...
1; # end of Underline.pm



Using exported 
symbols

Export variables
use Underline;
$underliner = "*";
print underlined( "WebGUI Conference 2004" );

WebGUI Conference 2004
****** ********** ****

Much nicer!



Objects
Objects are just blessed structures.

Inheritance is based on packages.



Constructors
Create a structure and bless it.

package Stooge;

sub new {
    my $class = shift;
    my $name = shift || die "Must pass name!";
    my $self = {
        name => $name,
        bonk_count => 0,
    };
    bless $self, $class;
    return $self;
}
    



Instantiating
Call the constructor

use Stooge;

my $larry = Stooge->new( "Larry" );
print Dumper( $larry );

$VAR1 = bless( {
                 'bonk_count' => 0,
                 'name' => 'Larry',
               }, 'Stooge' );



Methods
A method is a subroutine that expects a $self

package Stooge;

sub bonk {
    my $self = shift;
    ++$self->{bonks};
    print "Ow!\n";
}

sub bonk_count {
    my $self = shift;
    return $self->{bonks};
}



Calling methods
Call a method indirectly

my $curly = Stooge->new( 'cneal' );
$curly->bonk();
$curly->bonk();
print "Bonked ", $curly->bonk_count, " times\n";

Ow!
Ow!
Bonked 2 times



Use your accessors
Don't look at internals directly

print $stooge->{bonks}, " bonks"; # BAD

Keep interface to objects opaque
Objects should eat their own dog food

sub status {
    my $self = shift;

    print "I've got a headache!"
        if $self->bonk_count > 10;
}

Performance geeks say it's slower.  Ignore them.



Inheritance
Some classes may inherit from others

package Stooge::Moe;

our @ISA = qw( Stooge );

sub new {
    my $class = shift;
    my $self = $class->SUPER::new(@_);
    $self->{hairstyle} = 'bowl';
    return $self;
}

Uses parent's constructor, and adds its own stuff



Overriding methods
Create a new bonk method

sub bonk {
    my $self = shift;

    print "Why you...";
    $self->{bonk_count}++;
}

Instantiate with appropriate constructor
my $curly = Stooge->new( 'cneal' );
my $moe = Stooge::Moe->new( 'mhoward' );



Overridden methods
Perl checks the @ISA list to find method

$curly->bonk;  # Stooge::bonk
$moe->bonk;    # Stooge::Moe::bonk

$curly->bonk_count # Stooge::bonk_count
$moe->bonk_count   # Stooge::bonk_count

Method calls are determined at runtime
You can modify @ISA on the fly, but don't.



Where to Get More 
Information

• Learning Perl Objects, 
References & Modules, by 
Randal Schwartz

• Object-Oriented Perl, by 
Damian Conway



Bonus topics 
if time allows

map/grep

anonymous subroutines

Writing a module

Command-line wizardry

Handy magic variables

POD


